

平成26年12月6日

跡見学園女子大学
公開講座パソコンセミナー

「Excel 入門」
第1回 応用編

文学部 現代文化表現学科

准教授 伊藤穰

j-ito@atomi.ac.jp

<http://www2.mmc.atomi.ac.jp/~j-ito/>

目次

1.	論理関数 IF.....	1
2.	「IF」の概要.....	1
3.	論理式の種類.....	2
3.1.1.	等号.....	2
3.1.2.	不等号.....	2
4.	具体的な使い方.....	2
5.	ネスト.....	3
6.	複数の条件を記述.....	3
7.	COUNTIF.....	4
8.	COUNTIFS.....	5
9.	COUNTA、COUNTBLANK.....	6
10.	SUMIF.....	6
11.	SUMIFS.....	6
12.	VLOOKUP.....	7
13.	条件付き書式.....	9

※当テキストは Microsoft Excel 2010 を前提としていますが、他のバージョンとも大部分が共通しています。

1. 論理関数 IF

2. 「IF」の概要

IF という論理関数を使うと、ある条件に従って、セルに表示させる内容を変更することができます。たとえば、隣接するセルの値に応じて、表示する文字列や数値などを変えることができます。

IF は、以下のような構造になっています。論理式、真の場合、偽の場合の、三つの引数を持ちます。

```
=IF(論理式, [真の場合], [偽の場合])
```

[真の場合] とは、論理式が満たされる場合のことを意味します。

たとえば、セル A1 とセル A2 に、それぞれ“5”という数値が入っているとします。そして、論理式が「A1= A2」となっている場合、これは、論理式が満たされる（すなわち、正しい）ということになります。そのときには、「真の場合」が反映されることとなります。

関数の [真の場合] の部分には、真の場合に表示させたいものを記述します。ここには、セル番号や数値、文字列を記述できます。数式を記述することもできます。

文字列を記述する場合は、文字列をダブルクォート（キーボードの Shift キーを押しながら数字の 2 を押す）で括ります。

たとえば、論理式 A1=A2 が満たされたときに「正解です!」、満たされないときに「残念!」と表示させたい場合は、以下のように記述します。

```
=IF(A1 = A2, "正解です!", "残念!")
```

また、セル B1 の内容が「○」であった際に、セル A1 の内容を表示させ、そうでない場合は何も表示しない、という場合は、以下のように記述します。

```
=IF(B1 = "○", A1, "")
```

3. 論理式の種類

3.1.1. 等号

等号を用いると、特定のセルの数値が、ある数値と同じかどうかを調べることができます。

たとえば、セル A1 の数値が“7”なのか否かを調べる場合は、関数「IF」の論理式は「A1=7」と記述します。実際にセル A1 の数値が 7 であれば、真となります。

前述の例のように、セル同士の数値を比較することもできます。

3.1.2. 不等号

不等号を用いると、特定のセルの値が、ある数よりも大きい、あるいは小さいか否かを調べることができます。等号と組み合わせることで、ある数値以上、あるいは以下であるかを調べることができます。

たとえば、セル A1 が 10 以上であるかどうかを調べる場合には、「A1>=10」と記述します。実際にセル A1 の数値が 10 以上であれば、真となります。

A1 > B1	A1 は B1 より大きい
A1 >= B1	A1 は B1 以上
A1 < B1	A1 は B1 より小さい
A1 <= B1	A1 は B1 以下

4. 具体的な使い方

例として、次のような場面について説明します。

- セル A1 にテストの点数が入力されているものとする。
- そのテストは 60 点以上が合格で、59 点以下は不合格である。
- セル A2 に、関数「IF」を用いて、セル A1 が 60 点以上の場合には「合格」、59 点以下の場合には「不合格」と表示する。

この場合、関数「IF」は以下のように記述します。

```
=IF(A1 >= 60, "合格", "不合格")
```

5. ネスト

関数「IF」の [真の場合] や [偽の場合] の部分には、さらに関数「IF」を挿入することができます。このような構造をネストといいます。最大 7 階層までです。

例：A1 について、10 以上か、5 以上 10 未満か、5 未満かのいずれかを表示

```
=IF(A1 >= 10, "10 以上", IF(A1 >= 5, "5 以上 10 未満", "5 未満"))
```

6. 複数の条件を記述

論理式の部分に複数の条件を記述するには、and() や or() で囲み、コンマによって条件を区切っていきます。

複数の条件を同時に満たす場合は and、いずれかの条件を満たす場合は or を用います。以下の例では、女性で、かつ身長が 160cm 以上に該当する場合は「○」、そうでない場合は空白が表示されるようにしています。

	A	B	C	D	E	F
1		身長	性別	女性で160以上		
2	Aさん	167	女	=if(and(C2="女",B2>=160),"○","")		
3	Bさん	151	女			
4	Cさん	175	男			
5	Dさん	181	男			
6	Eさん	161	女			

図 1.1 : and による複数条件

以下のように記述しています。この例では 2 行目と 6 行目に○が表示されます。

```
=if( and(C2 = "女", B2 >= 160), "○", "")
```

コラム：セル内の空白について

論理式や、検索条件においてセル内の文字列を対象とするときに、セル内に空白が入っていることによって、一致しないと判定されることがあります。これを防ぐためには、セル内の空白を、置換の機能によって全て削除しておくことが考えられます。

7.COUNTIF

ある範囲について、条件を満たすセルの個数をカウントするには、COUNTIF を使うと便利です。

```
=COUNTIF (範囲, 検索条件)
```

たとえば、セル B2 から B10 までについて、「○」が入力されているセルの個数を数えるには、以下のように記述します。

```
=COUNTIF (B2:B10, "○")
```

この場合、検索条件が、完全に一致する場合のみカウントされます。ですから、セルの中に「○」以外の文字がある場合は、以下のようにワイルドカード「*」（アスタリスク）を使います。

ワイルドカードは、文字列の前、後のどちらにもつけることができます。

```
=COUNTIF (B2:B10, "○*")
```

また、検索条件の部分でセルを参照している場合は、「&」記号を使ってワイルドカードと連結します。

```
=COUNTIF (B2:B10, "*" & B12 )
```

検索条件には数式を記述することもできます。数式は、ダブルクォートで囲みます。以下の例では、セル C2 からセル C10 までについて、数値が 60 以上のセルの個数をカウントしています。

```
=COUNTIF (C2:C10, ">=60")
```

8.COUNTIFS

複数の条件で検索したい場合は、COUNTIFS を使うと便利です (2007 以降)。以下の例では、頭数が 3 頭以上で、かつ、区画が A である行の個数をカウントすることができます。

	A	B	C	D	E
1		頭数	区画		
2	いぬ	3	A		
3	ねこ	4	B		
4	さる	4	A		
5	パンダ	1	B		
6	サメ	3	F		
7	キリン	0	A		
8	ライオン	4	B		
9					
10	検索条件:				
11	頭数(以上)	3			
12	区画	A			
13	個数	=COUNTIFS(B2:B8,">="&B11,C2:C8,B12)			

図 3.1 : COUNTIFS の例

この例では数式部分でセルを参照しているので、「&」記号で連結しています。

コラム：重複するセルの発見

内容が重複しているセルを発見するには、COUNTIF により、「自分よりも上のセルに、自分と同じものがいくつ含まれるか」をカウントします。

以下の例では、2 行目から自分の行までについて、自分自身と同じ内容をカウントしていますので、重複すると、結果は 2 と表示されます。

	A	B	C	D	E
1		頭数	区画	重複	
2	いぬ	3	A	1	
3	ねこ	4	B	1	
4	さる	4	A	1	
5	いぬ	3	A	2	
6	パンダ	1	B	=COUNTIF(\$A\$2:A6,A6)	
7	サメ	3	F	1	
8	キリン	0	A	1	

9. COUNTA、COUNTBLANK

文字列や数値が入力されているセルの個数をカウントするには、COUNTA を用います。

似た関数として COUNT がありますが、そちらは数値が入力されたセルだけがカウントされます。

```
=COUNTA(範囲)
```

また、空白のセルの個数をカウントするには、COUNTBLANK を用います。

```
=COUNTBLANK(範囲)
```

10. SUMIF

検索条件に合うセルを抽出して合計するには、SUMIF を使います。

```
=SUMIF(範囲, 検索条件, 合計範囲)
```

たとえば、前頁の図 3.1 の表において、区画 A のものの頭数の合計を出したいときは、以下のように記述します。

```
=SUMIF(C2:C8, "A", B2:B8)
```

この記述では、セル C2 からセル C8 までについて、「A」が入力されている行についてのみ、セル B2 からセル B8 までの数値の合計を計算しています。

11. SUMIFS

複数の検索条件によってセルを抽出して合計するには、SUMIFS を使います。

```
=SUMIFS(合計対象範囲, 条件範囲1, 条件1, 条件範囲2, 条件2, ...)
```


たとえば、ある一定の日付の範囲についてのみ計算をさせたい場合には、条件 1 を開始の日付、条件 2 を終了の日付とします。

以下の例では、セル F8 において、1 月 7 日から 1 月 10 日までの果物の合計を表示しています。

F8 fx =SUMIFS(C2:C11,A2:A11,E8,A2:A11,E9)						
	A	B	C	D	E	F
1	日付	果物	数量			
2	1月5日	りんご	20		りんご	40
3	1月5日	いちご	16		みかん	24
4	1月7日	いちご	16		いちご	48
5	1月7日	バナナ	30		バナナ	30
6	1月10日	みかん	24			
7	1月10日	りんご	20		条件	合計
8	1月10日	いちご	16		>=2011/1/7	106
9	1月12日	みかん	24		<=2011/1/10	
10	1月12日	バナナ	10			
11	1月13日	いちご	32			

図 6.1 : SUMIFS の活用

セル F8 には、以下のように記述しています。

```
=SUMIFS(C2:C11, A2:A11, E8, A2:A11, E9)
```

セル E8 とセル E9 には、日付の範囲をあらかじめ記述しておきます。

合計したいのはセル C2 からセル C11 なので、この部分を合計対象範囲として指定しています。そして、日付について、条件にあうものだけを計算します。

1 2 . VLOOKUP

VLOOKUP を使うと、表の中から、ある条件を満たすセルが含まれた行を見つけ出し、その行の特定の列のセルを抽出することができます。

```
=VLOOKUP(検索値, 範囲, 列, 検索の型)
```

「列」は、選択した範囲において、何列目なのかを記述します。「検索の型」には、完全一致の場合は 0、近似の場合は 1 を記述します。通常は 0 で良いでしょう。

たとえば、以下の表の B 列に、コード表からコードと動物名の対応関係を見つけ出し、それを表示させることにします。

	A	B	C	D	E	F	G
1	区画表					コード表	
2	コード	動物名	区画			コード	動物名
3	M801					S101	いぬ
4	M303					L50	キリン
5	S316					W1001	サメ
6	S203					S203	さる
7	W1001					S316	ねこ
8	L50					M303	パンダ
9	S101					M801	ライオン

図 7.1 : VLOOKUP の例

その場合、以下のように記述します。

```
=VLOOKUP(A3, $F$3:$G$9, 2, 0)
```

範囲は、オートフィルによって他のセルにコピーした際に参照先が移動しないように、絶対参照にしています。

範囲の 2 列目を取得したいので、列には 2 と記述しています。

ところで、範囲の中に検索値が存在しない場合は、エラーとして「#N/A」と表示されます。これを避けるには、IFERROR で囲ってやる方法が考えられます。

```
=IFERROR(表示する値, エラーの際に表示する値)
```

具体的には、以下のように記述します。検索値が存在しない場合は、「該当なし」と表示させています。

```
=IFERROR(VLOOKUP(A3, $F$3:$G$9, 2, 0), "該当なし")
```

1.3. 条件付き書式

条件付き書式を設定するには、[ホーム] タブの [スタイル] グループにある [条件付き書式] をクリックします。

条件付き書式には、あらかじめ用意されたパターンがあり、その中から選択することで、簡単に設定することができます。

また、[新しいルール] を選択すると、任意のルールに基づいて、セルに条件付き書式を設定することができます。




図 8.1 : 条件付き書式




図 8.2 : 新しい書式ルールの設定

このダイアログで、ルールの種類を選択し、ルールの内容を編集することで、ルールを作成することができます。

セルに設定したルールをクリアしたい場合は、[条件付き書式] をクリックして、[ルールのクリア] をクリックします。